Clipper circuits
A circuit which removes the peak of a waveform is known as a clipper. A negative
clipper is shown in Figure below. This schematic diagram was produced with Xcircuit
schematic capture program. Xcircuit produced the SPICE net list Figure below,
except for the second, and next to last pair of lines which were inserted with a text
editor.

*SPICE 03437.eps
* A K ModelName
D1 0 2 diode
R1 2 1 1.0k
V1 1 0 SIN(0 5 1k)
.model diode d
.tran .05m 3m
.end

Clipper: clips negative peak at -0.7 V.
During the positive half cycle of the 5 V peak input, the diode is reversed biased.
The diode does not conduct. It is as if the diode were not there. The positive half
cycle is unchanged at the output V(2) in Figure below. Since the output positive
peaks actually overlays the input sinewave V(1), the input has been shifted upward
in the plot for clarity. In Nutmeg, the SPICE display module, the command “plot
v(1)+1)” accomplishes this.

V(1)+1 is actually V(1), a 10 Vptp sinewave, offset by 1 V for display clarity. V(2)
output is clipped at -0.7 V, by diode D1.
During the negative half cycle of sinewave input of Figure above, the diode is
forward biased, that is, conducting. The negative half cycle of the sinewave is
shorted out. The negative half cycle of V(2) would be clipped at 0 V for an ideal
diode. The waveform is clipped at -0.7 V due to the forward voltage drop of the
silicon diode. The spice model defaults to 0.7 V unless parameters in the model
statement specify otherwise. Germanium or Schottky diodes clip at lower voltages.
Closer examination of the negative clipped peak (Figure above) reveals that it
follows the input for a slight period of time while the sinewave is moving toward 0.7 V. The clipping action is only effective after the input sinewave exceeds -0.7 V.
The diode is not conducting for the complete half cycle, though, during most of it.
The addition of an anti-parallel diode to the existing diode in Figure above yields
the symmetrical clipper in Figure below.

*SPICE 03438.eps
D1 0 2 diode
D2 2 0 diode
R1 2 1 1.0k
V1 1 0 SIN(0 5 1k)
.model diode d
.tran 0.05m 3m
.end

Symmetrical clipper: Anti-parallel diodes clip both positive and negative peak, leaving a ± 0.7 V
output.
Diode D1 clips the negative peak at -0.7 V as before. The additional diode D2
conducts for positive half cycles of the sine wave as it exceeds 0.7 V, the forward
diode drop. The remainder of the voltage drops across the series resistor. Thus,
both peaks of the input sinewave are clipped in Figure below. The net list is in
Figure above

Diode D1 clips at -0.7 V as it conducts during negative peaks. D2 conducts for
positive peaks, clipping at 0.7V.
The most general form of the diode clipper is shown in Figure below. For an ideal
diode, the clipping occurs at the level of the clipping voltage, V1 and V2. However,
the voltage sources have been adjusted to account for the 0.7 V forward drop of
the real silicon diodes. D1 clips at 1.3V +0.7V=2.0V when the diode begins to
conduct. D2 clips at -2.3V -0.7V=-3.0V when D2 conducts.

*SPICE 03439.eps
V1 3 0 1.3
V2 4 0 -2.3
D1 2 3 diode
D2 4 2 diode
R1 2 1 1.0k
V3 1 0 SIN(0 5 1k)
.model diode d
.tran 0.05m 3m
.end

D1 clips the input sinewave at 2V. D2 clips at -3V.
The clipper in Figure above does not have to clip both levels. To clip at one level
with one diode and one voltage source, remove the other diode and source.
The net list is in Figure above. The waveforms in Figure below show the clipping of
v(1) at output v(2).

D1 clips the sinewave at 2V. D2 clips at -3V.
There is also a zener diode clipper circuit in the “Zener diode” section. A zener
diode replaces both the diode and the DC voltage source.
A practical application of a clipper is to prevent an amplified speech signal from
overdriving a radio transmitter in Figure below. Over driving the transmitter
generates spurious radio signals which causes interference with other stations. The
clipper is a protective measure.

Clipper prevents over driving radio transmitter by voice peaks.
A sinewave may be squared up by overdriving a clipper. Another clipper application
is the protection of exposed inputs of integrated circuits. The input of the IC is
connected to a pair of diodes as at node “2” of Figure above . The voltage sources
are replaced by the power supply rails of the IC. For example, CMOS IC's use 0V
and +5 V. Analog amplifiers might use ±12V for the V1 and V2 sources.






REVIEW
A resistor and diode driven by an AC voltage source clips the signal
observed across the diode.
A pair of anti-parallel Si diodes clip symmetrically at ±0.7V
The grounded end of a clipper diode(s) can be disconnected and wired to a
DC voltage to clip at an arbitrary level.
A clipper can serve as a protective measure, preventing a signal from
exceeding the clip limits.

Source: http://www.allaboutcircuits.com/vol_3/chpt_3/6.html

